

# LOS ESFUERZOS MECANICOS EN CONDUCTORES DE LINEAS ELECTRICAS AEREAS DE ALTA TENSION, RESUELTOS EN BASIC

Por Pedro José Martínez-Lacañina

## INTRODUCCION

La expresión algebraica en la que se fundamenta el estudio de los esfuerzos mecánicos en conductores de líneas eléctricas aéreas es la Ecuación de Cambio de Condiciones. Esta ecuación es de tercer grado y se resuelve, en la práctica, por el método interactivo de Newton.

Lógicamente la resolución de esta ecuación ya es conocida, informáticamente hablando, por las grandes empresas suministradoras e instaladoras. Pero también es cierto que muchas de las pequeñas y medianas empresas instaladoras, así como la casi totalidad de los profesionales en el ejercicio libre, no se aprovechan de las posibilidades informáticas que en este campo están a su alcance, a pesar de que dicha resolución está considerada como pesada y engorrosa, ya que la ecuación se resuelve por un método de aproximación por tanteo, y la mayoría de las veces la búsqueda de la solución da la sensación de ser interminable.

Afianzado por el auge que ha adquirido la aplicación de los ordenadores personales a todos los aspectos de la técnica, las anteriores consideraciones me han llevado a

confeccionar un programa, en lenguaje BASIC, en el que sólo con la introducción de las características del conductor, y del tendido, se obtienen los resultados de la ecuación para todas las hipótesis exigidas en la normativa vigente al respecto.

## 1. DESCRIPCION DEL PROGRAMA

### 1.1. DESCRIPCION DEL ORDINOGRAMA

Se ha creído conveniente presentar primero un ordinograma general (fig. núm. 1), en el que se exponen todas las fases o procesos que se han de realizar, acompañándolo de otros (fig. núm. 2, 3, 4 y 5), en los que se desglosan cada uno de estos procesos por el hecho de facilitar su comprensión.

En la figura núm. 1 podemos ver que una vez realizada la entrada de datos, se pasa al proceso (A), comprendido entre las anotaciones (A-A'). Este proceso se halla desglosado en la figura núm. 2.

Al estudiar la figura núm. 2 se aprecia que el mismo proceso se

encarga de elegir la hipótesis de tracción máxima admisible, en función de la altitud de la zona donde se realice el tendido para acto seguido pasar a realizar el cálculo de los parámetros correspondientes a la citada hipótesis.

El siguiente proceso (B) se encuentra comprendido entre las anotaciones (B-B') del ordinograma general, su desglose se expone en las figuras núms. 3 y 4. Se ha optado por separar este proceso en dos figuras, porque consta de dos partes claramente diferenciadas.

La primera parte (B-1), detallada en la figura núm. 3, corresponde a la determinación de los parámetros de cada una de las hipótesis de flechas máximas.

La segunda parte (1-B'), del proceso (B), está detallada en la figura núm. 4; corresponde a la exposición pormenorizada de todos los pasos a seguir para la resolución de la ecuación de cambio de condiciones en cada una de las hipótesis antes citada, según la hipótesis de tracción admisible elegida en el proceso anterior (A).

Por último, el proceso (C) comprendido entre las anotaciones (C-C') se desglosa en la figura núm. 5.

En este proceso se recogen los valores obtenidos en la resolución de la ecuación de cambio de condiciones, y se dan las órdenes oportunas para su escritura.

## 1.2. ACLARACIONES AL PROGRAMA:

En este apartado se pasa a enumerar la concordancia de las instrucciones del programa que, como se indicó al principio, se ha realizado en BASIC con cada uno de los procesos que componen el ordinograma.

— Entrada de datos: Instrucciones 10 a 500.

— Proceso (A): Instrucciones 510 a 1.190.

— Proceso (B-1): Instrucciones 1.200 a 2.060.

— Proceso (1-B'): Instrucciones 2.070 a 2.270.

— Proceso (C): Instrucciones 2.280 a 2.770.

— Disyuntiva de seguir o acabar: Instrucciones 2.780 a 3.010.

— Las instrucciones 3.020 a 3.670 corresponden a instrucciones que se repiten en varias fases del desarrollo del programa, y que se han colocado al final para reclamarlas en los casos que sea necesaria su intervención.

Por otro lado, para un correcto seguimiento del programa, hay que tener en cuenta lo siguiente:

— Las variables AC, BC y CC son contadores auxiliares al desarrollo del programa. Al contador AC se le ha puesto el tope del valor ( $1 \times 10^6$ ), para fijar el número de decimales a obtener en el cálculo de las tracciones en las hipótesis de flechas máximas, a los otros contadores se le ha puesto el tope del valor tres, por ser ese el número de hipótesis de flechas máximas.

— Las variables R1, R2, R3 y S1, S2, S3 se han utilizado para reservar

zonas de memoria en las cuales se almacenan los valores de los coeficientes de sobrecarga para cada una de las tres hipótesis de flechas máximas, y tracciones máximas obtenidas en la resolución de la ecuación de cambio de condiciones respectivamente.

— Las variables X1 y X2 corresponden a la asignación de los valores de las temperaturas a considerar en

las hipótesis de tracción máxima admisible y de flechas máximas respectivamente.

Para dar por finalizada esta somera memoria explicativa, simplemente puntualizar que el programa se ha realizado en lenguaje BASIC estándar, para facilitar su traducción a los diferentes dialectos, que actualmente existen en el mercado. ■




FIGURA NUM. 1



- DETALLE DEL PROCESO (A) -




FIGURA NUM. 2

- DETALLE DEL PROCESO (B) -





FIGURA NUM. 3

FIGURA NUM. 4



- DETALLE DEL PROCESO (C) -




FIGURA NUM. 5

```

10 REM --- Pedro J. Martinez Lacañina ---
20 CLS:PRINT:PRINT
30 REM --- PRESENTACION DE LA E. C. C. ---
40 Y$="P R E S E N T A C I O N":X=LEN(Y$)
50 Y=INT((80-X)/2):X$=STRING$(80,42)
60 PRINT X$;SPC(Y);Y$;SPC(Y);X$
70 PRINT " -La ecuacion a emplear es:"
80 PRINT:PRINT
90 PRINT SPC(30);"T2^2*(T2+J+K)=I"
100 PRINT:PRINT:PRINT
110 PRINT SPC(10);"J=(P0^2*M1^2*V^2*E)/(24*T1^2)"
120 PRINT
130 PRINT SPC(10);"K=AL*(X2-X1)*E-T1"
140 PRINT
150 PRINT SPC(10);"I=P0^2*M2^2*V^2*E/24"
160 PRINT:PRINT
170 GOSUB 3020
180 CLS:PRINT:PRINT
190 REM --- FASE DE ENTRADA DE DATOS ---
200 Y$=" ENTRE LOS DATOS DEL CONDUCTOR "
210 X=LEN(Y$):Y=INT((80-X)/2)
220 Z$=STRING$(Y,42)
230 PRINT Z$;Y$;Z$
240 PRINT:PRINT:PRINT
250 INPUT " -Peso en (KG/M):";P
260 INPUT " -Coef. de dil. lineal en (1/C):";AL
270 INPUT " -Modulo elástico en (KG/MM2):";E
280 INPUT " -Diámetro en (MM):";D
290 INPUT " -Sección en (MM2):";S
300 INPUT " -Tensión de rotura en (KG):";TR
310 INPUT " -Coef. de seguridad a adoptar:";CS
320 IF NOT CS<3 THEN GOTO 350
330 PRINT " (Le recomiendo que para cumplir la
normativa vigente adopte 3 como va-
lor minimo)"
340 LOCATE 13,1:GOTO 310
350 TM=TR/3:T1=TM/S:P0=P/S
360 CLS:PRINT:PRINT
370 Y$=" ENTRE LOS DATOS DEL TENDIDO "
380 X=LEN(Y$):Y=INT((80-X)/2)
390 PRINT Z$;Y$;Z$
400 PRINT:PRINT:PRINT
410 INPUT " -Vano en (M):";V
420 INPUT " -Altitud en (M) de
la zona del tendido:";H
430 Y$="D A T O S D E E N T R A D A"
440 X=LEN(Y$):Y=INT((80-X)/2)
450 LPRINT:LPRINT:LPRINT
460 LPRINT X$;SPC(Y);Y$;
470 LPRINT SPC(Y);X$
480 LPRINT "P=";P;"(KG/M)",,"AL=";AL;"(1/C)",
"E=";E;"(KG/MM2)",,"D=";D;"(MM)", "S=";S;
"(MM2)",,"TR=";TR;"(KG)", "CS=";CS
490 LPRINT "V=";V;"(M)",,"H=";H;"(M)"
500 LPRINT "T1=TR/S*CS=";T1;"(KG/MM2)",,"
"PO=P/S=";PO;"(KG/M*MM2)"

```

```

510 IF 500>H THEN A=-5:X1=A:GOTO 530
520 IF 500<=H AND H<=1000 THEN

        B=-15:X1=B:GOTO 910

        ELSE C=-20:X1=C:GOTO 1060
530 CLS:PRINT:PRINT
540 REM ---- Determinación de los valores en la
        hipótesis de tracción máxima admi-
        sible en zona (A) ----
550 Y$="T E N D I D O   E N   Z O N A   (A)"
560 X=LEN(Y$):Y=INT((80-X)/2)
570 PRINT X$;SPC(Y);Y$;SPC(Y);X$
580 GOSUB 3040
590 PV$="      -Presión de viento:"
600 FV$="      -Fuerza del viento:"
610 GOSUB 3160
620 IF NOT D<=16 THEN GOTO 690
630 PV=60:PRINT:PRINT
640 PRINT PV$;TAB(46);
650 PRINT "60.00      (KG/M2)"
660 LPRINT:LPRINT:LPRINT
670 LPRINT PV$;TAB(46);
680 LPRINT "60.00      (KG/M2)":GOTO 750
690 PV=50:PRINT:PRINT
700 PRINT PV$;TAB(46);
710 PRINT "50.00      (KG/M2)"
720 LPRINT:LPRINT:LPRINT
730 LPRINT PV$;TAB(46);
740 LPRINT "50.00      (KG/M2)"
750 FV=PV*(D/1000)
760 PRINT FV$;TAB(45);
770 PRINT USING U1$;FV;
780 PRINT "      (KG/M)":PRINT:PRINT
790 LPRINT FV$;TAB(45);
800 LPRINT USING U1$;FV;
810 LPRINT "      (KG/M)":LPRINT:LPRINT
820 ER=SQR(FV^2+P^2):M1=ER/P
830 GOSUB 3290
840 PRINT X1$;
850 PRINT TAB(46);A;"      (C)"
860 LPRINT X1$;
870 LPRINT TAB(46);A;"      (C)"
880 PRINT:PRINT
890 GOSUB 3020
900 GOTO 1200
910 CLS:PRINT:PRINT
920 REM ---- Determinación de los valores en la
        hipótesis de tracción máxima admi-
        sible en zona (B) ----
930 Y$="T E N D I D O   E N   Z O N A   (B)"
940 X=LEN(Y$):Y=INT((80-X)/2)
950 PRINT X$;SPC(Y);Y$;SPC(Y);X$
960 SH=(180*SQR(D))/1000
970 ER=SH+P:M1=ER/P
980 GOSUB 3040:GOSUB 3150:GOSUB 3210

```



```

990 PRINT X1$;
1000 PRINT TAB(45);B;"          (C)"
1010 LPRINT X1$;
1020 LPRINT TAB(45);B;"          (C)"
1030 PRINT:PRINT
1040 GOSUB 3020
1050 GOTO 1200
1060 CLS:PRINT:PRINT
1070 REM ---- Determinación de los valores en la
          hipótesis de tracción máxima admi-
          sible en zona (C) ----
1080 Y$="T E N D I D O   E N   Z O N A   (C)"
1090 X=LEN(Y$):Y=INT((80-X)/2)
1100 PRINT X$;SPC(Y);Y$;SPC(Y);X$
1110 SH=(360*SQR(D))/1000
1120 ER=SH+P:M1=ER/P
1130 GOSUB 3040:GOSUB 3150:GOSUB 3210
1140 PRINT X1$;
1150 PRINT TAB(45);C;"          (C)"
1160 LPRINT X1$;
1170 LPRINT TAB(45);C;"          (C)"
1180 PRINT:PRINT
1190 GOSUB 3020
1200 CLS:PRINT:PRINT
1210 REM ---- EXPOSICION DE LAS HIPOTESIS
          DE FLECHAS MAXIMAS ----
1220 P$=" (Valores de los parámetros para que"
1230 Q$=" se presenten las hipótesis de flechas"
1240 O$=" maximas)"
1250 PRINT P$;Q$;O$
1260 PRINT STRING$(80,45)
1270 LPRINT:LPRINT
1280 LPRINT P$;Q$;O$
1290 LPRINT STRING$(80,45)
1300 BC=0
1310 BC=BC+1
1320 IF BC=2 THEN GOTO 1650
1330 IF BC=3 THEN GOTO 1790
1340 REM ---- 1.- HIPOTESIS DE VIENTO. ----
1350 PV$=" -Presión de viento:"
1360 FV$=" -Fuerza del viento:"
1370 GOSUB 3460
1380 PRINT " - V I E N T O .-
          ===== "
1390 LPRINT " - V I E N T O .-
          ===== "
1400 IF NOT D<=16 THEN GOTO 1480
1410 PV=60:PRINT:PRINT:PRINT
1420 PRINT PV$;TAB(46);
1430 PRINT "60.00          (KG/M2)"
1440 LPRINT:LPRINT
1450 LPRINT PV$;TAB(46);
1460 LPRINT "60.00          (KG/M2)"
1470 GOTO 1540
1480 PV=50:PRINT:PRINT:PRINT
1490 PRINT PV$;TAB(46);
1500 PRINT "50.00          (KG/M2)"
1510 LPRINT:LPRINT
1520 LPRINT PV$;TAB(46);
1530 LPRINT "50.00          (KG/M2)"

```

```

1540 FV=PV*(D/1000)
1550 PRINT FV$;TAB(45);
1560 PRINT USING U1$;FV;
1570 PRINT " (KG/M)"
1580 LPRINT FV$;TAB(45);
1590 LPRINT USING U1$;FV;
1600 LPRINT " (KG/M)"
1610 ER=SQR(FV^2+P^2)
1620 M2=ER/P:X2=15
1630 GOSUB 3500:GOSUB 3630
1640 GOTO 2050
1650 CLS:PRINT:PRINT
1660 REM ---- 2.- HIPOTESIS DE TEMPERATURA. ----
1670 PRINT " - T E M P E R A T U R A .-
          ====="
1680 LPRINT:LPRINT
1690 LPRINT " - T E M P E R A T U R A .-
          ====="
1700 S0$=" -Sin SOBRECARGA ."
1710 GOSUB 3460
1720 PRINT:PRINT:PRINT
1730 PRINT S0$
1740 LPRINT:LPRINT
1750 LPRINT S0$
1760 ER=P:M2=ER/P:X2=50
1770 GOSUB 3500:GOSUB 3630
1780 GOTO 2050
1790 CLS:PRINT:PRINT
1800 REM ---- 3.- HIPOTESIS DE HIELO. ----
1810 PRINT " - H I E L O .-
          ===== "
1820 LPRINT:LPRINT
1830 LPRINT " - H I E L O .-
          ===== "
1840 X2=0
1850 IF 500>H THEN
          SH=0:GOTO 1890
1860 IF 500<=H AND H<=1000 THEN
          GOTO 1870
          ELSE GOTO 1880
1870 SH=(180*SQR(D))/1000:GOTO 1890
1880 SH=(360*SQR(D))/1000
1890 ER=SH+P:M2=ER/P
1900 SH$=" -Peso del manguito de hielo:"
1910 GOSUB 3460
1920 PRINT:PRINT:PRINT
1930 PRINT SH$;TAB(45);
1940 PRINT USING U2$;SH;
1950 PRINT " (KG/M)"
1960 LPRINT:LPRINT
1970 LPRINT SH$;TAB(45);
1980 LPRINT USING U2$;SH;
1990 LPRINT " (KG/M)"
2000 GOSUB 3500

```



```

2010 PRINT X2$;
2020 PRINT TAB(46);X2;"      (C)"
2030 LPRINT X2$;
2040 LPRINT TAB(46);X2;"      (C)"
2050 PRINT:PRINT
2060 GOSUB 3020
2070 CLS:PRINT:PRINT
2080 REM ---- FASE DE CALCULO ----
2090 Y$="F A S E      D E      C A L C U L O"
2100 X=LEN(Y$):Y=INT((80-X)/2)
2110 PRINT X$;SPC(Y);Y$;SPC(Y);X$
2120 I=PO^2*M2^2*V^2*E/24
2130 J=(PO^2*M1^2*V^2*E)/(24*T1^2)
2140 K=AL*(X2-X1)*E-T1
2150 T2=0:AC=0
2160 LOCATE 15,1:PRINT SPC(21);
2170 PRINT "ESTOY RESOLVIENDO LA ECUACION"
2180 LET L=T2^2*(T2+J+K)
2190 IF AC>=1 THEN GOTO 2220
2200 IF I-L>=1 THEN T2=T2+1:GOTO 2180
2210 T2=T2-1:AC=1:GOTO 2180
2220 IF AC=1000000! THEN GOTO 2250
2230 IF 0<I-L THEN T2=T2+(.1/AC):GOTO 2180
2240 T2=T2-(.1/AC):AC=AC*10:GOTO 2180
2250 IF BC=1 THEN S1=T2:R1=M2:GOTO 1310
2260 IF BC=2 THEN S2=T2:R2=M2:GOTO 1310
2270 S3=T2:R3=M2
2280 CLS:PRINT:PRINT
2290 REM ---- EXPOSICION DE RESULTADOS ----
2300 Y$="RESULTADOS DE LA E.C.C."
2310 X=LEN(Y$):Y=INT((80-X)/2)
2320 PRINT X$;SPC(Y);Y$;SPC(Y);X$
2330 LPRINT:LPRINT
2340 LPRINT SPC(Y);Y$;SPC(Y);
2350 LPRINT STRING$(80,61):LPRINT
2360 CC=0
2370 CC=CC+1
2380 IF CC=2 THEN GOTO 2450
2390 IF CC=3 THEN GOTO 2500
2400 T2=S1:M2=R1:PRINT
2410 PRINT "      - HIPOTESIS DE VIENTO:
      =====":PRINT
2420 LPRINT:LPRINT
2430 LPRINT "      - HIPOTESIS DE VIENTO:
      ====="
2440 LPRINT:GOTO 2540
2450 T2=S2:M2=R2:PRINT
2460 PRINT "      - HIPOTESIS DE TEMPERATURA:
      =====":PRINT
2470 LPRINT:LPRINT
2480 LPRINT "      - HIPOTESIS DE TEMPERATURA:
      ====="
2490 LPRINT:GOTO 2540
2500 T2=S3:M2=R3:PRINT
2510 PRINT "      - HIPOTESIS DE HIELO:
      =====":PRINT
2520 LPRINT:LPRINT
2530 LPRINT "      - HIPOTESIS DE HIELO:
      =====":LPRINT

```

```

2540 PRINT "      *- T2 =" ;TAB(43);
2550 PRINT USING U2$;T2;
2560 PRINT " (KG/MM2)"
2570 LPRINT "      *- T2 =" ;TAB(43);
2580 LPRINT USING U2$;T2;
2590 LPRINT " (KG/MM2)"
2600 PRINT "      *- T2*S=" ;TAB(40);
2610 PRINT USING U3$;T2*S;
2620 PRINT " (KG)"
2630 LPRINT "      *- T2*S=" ;TAB(40);
2640 LPRINT USING U3$;T2*S;
2650 LPRINT " (KG)"
2660 F=(PO*M2*(V^2))/(8*T2)
2670 PRINT "      *- F =" ;TAB(43);
2680 PRINT USING U2$;F;
2690 PRINT " (M)"
2700 LPRINT "      *- F =" ;TAB(43);
2710 LPRINT USING U2$;F;
2720 LPRINT " (M)"
2730 PRINT:PRINT
2740 GOSUB 3020
2750 CLS:PRINT:PRINT
2760 IF CC=3 THEN GOTO 2780
2770 GOTO 2370
2780 REM ---- CONCLUSION O REINICIO ----
2790 Y$="CUADRO FINAL DE OPCIONES"
2800 X=LEN(Y$):Y=INT((80-X)/2)
2810 PRINT X$;SPC(Y);Y$;SPC(Y);X$
2820 PRINT:PRINT:PRINT:PRINT
2830 PRINT "1.- S E G U I R.
           2.- A C A B A R.":PRINT:PRINT
2840 INPUT " - Elija una opción:";Z
2850 PRINT:PRINT:PRINT
2860 IF Z<>1 AND Z<>2 THEN LOCATE 15,1:GOTO 2840
2870 ON Z GOTO 2880,2950
2880 LOCATE 11,1
2890 PRINT "1.- CALCULOS CON OTRO CONDUCTOR.
           2.- CALCULOS CON EL MISMO CONDUCTOR."
2900 PRINT:PRINT
2910 INPUT " - Elija una opción:";W
2920 PRINT:PRINT:PRINT
2930 IF W<>1 AND W<>2 THEN LOCATE 15,1:GOTO 2910
2940 ON W GOTO 180, 360
2950 CLS:PRINT:PRINT
2960 PRINT:PRINT:PRINT:PRINT
2970 Y$="FIN SESION DE TRABAJO"
2980 X=LEN(Y$):Y=INT((80-X)/2)
2990 PRINT X$;SPC(Y);Y$;SPC(Y);X$
3000 LPRINT:LPRINT:LPRINT
3010 LPRINT STRING$(Y,61);Y$;STRING$(Y,61):END
3020 INPUT " Pulse (ENTER) para continuar. ";A$
3030 RETURN
3040 M$=" (Valores de los parámetros para que"
3050 N$=" se presente la hipótesis de tracción"
3060 O$=" maxima admisible)"
3070 PRINT M$;N$;O$:PRINT STRING$(80,45)
3080 LPRINT:LPRINT:LPRINT
3090 LPRINT X$;SPC(Y);Y$;SPC(Y);X$

```

```

3100 LPRINT M$;N$;O$:LPRINT STRING$(80,45)
3110 U1$="###.###"
3120 U2$="###.#####"
3130 U3$="#####.#####"
3140 RETURN
3150 SH$="      -Peso del manguito de hielo:"
3160 ER$="      *- ESFUERZO TOTAL RESULTANTE="
3170 M1$="      *- COEF. DE SOBRECARGA (M1)="
3180 T1$="      *- TRACCION MAXIMA ADMISIBLE (T1)="
3190 X1$="      *- TEMPERATURA A CONSIDERAR (X1)="
3200 RETURN
3210 PRINT:PRINT:PRINT
3220 PRINT SH$;TAB(45);
3230 PRINT USING U2$;SH;
3240 PRINT " (KG/M)":PRINT:PRINT
3250 LPRINT:LPRINT:LPRINT
3260 LPRINT SH$;TAB(45);
3270 LPRINT USING U2$;SH;
3280 LPRINT. " (KG/M)":LPRINT:LPRINT
3290 PRINT:PRINT ER$;TAB(45);
3300 PRINT USING U2$;ER;
3310 PRINT " (KG/M)"
3320 LPRINT:LPRINT ER$;TAB(45);
3330 LPRINT USING U2$;ER;
3340 LPRINT " (KG/M)"
3350 PRINT M1$;TAB(45);
3360 PRINT USING U2$;M1
3370 LPRINT M1$;TAB(45);
3380 LPRINT USING U2$;M1
3390 PRINT T1$;TAB(45);
3400 PRINT USING U2$;T1;
3410 PRINT " (KG/MM2)"
3420 LPRINT T1$;TAB(45);
3430 LPRINT USING U2$;T1;
3440 LPRINT " (KG/MM2)"
3450 RETURN
3460 ER$="      *- ESFUERZO TOTAL RESULTANTE="
3470 M2$="      *- COEF. DE SOBRECARGA (M2)="
3480 X2$="      *- TEMPERATURA A CONSIDERAR (X2)="
3490 RETURN
3500 PRINT:PRINT
3510 PRINT ER$;TAB(45);
3520 PRINT USING U2$;ER;
3530 PRINT " (KG/M)"
3540 LPRINT:LPRINT:LPRINT
3550 LPRINT ER$;TAB(45);
3560 LPRINT USING U2$;ER;
3570 LPRINT " (KG/M)"
3580 PRINT M2$;TAB(45);
3590 PRINT USING U2$;M2
3600 LPRINT M2$;TAB(45);
3610 LPRINT USING U2$;M2
3620 RETURN
3630 PRINT X2$;
3640 PRINT TAB(45);X2;"          (C)"
3650 LPRINT X2$;
3660 LPRINT TAB(45);X2;"          (C)"
3670 RETURN

```



## PEDRO JOSE MARTINEZ LACÑINA

### ESTUDIOS Y CONOCIMIENTOS

— Ingeniería Técnica Industrial: *Realizados los estudios en la Escuela Universitaria de Ingeniería Técnica Industrial de Sevilla.*

*Años 1975 a 1978, realizada la especialidad de electricidad, sección de CENTRALES Y REDES ELECTRICAS».*

*Años 1985 a 1986, realizada la especialidad de electricidad, sección «MAQUINAS ELECTRICAS».*

— Diplomado por la Universidad Internacional Menéndez Pelayo en Energías Renovables.

— Curso de Aparata Eléctrica, organizado por el Departamento de Ingeniería Eléctrica de la Universidad de Sevilla.

— Curso sobre Cálculo de Líneas Eléctricas Aéreas de Alta Tensión, organizado por el Colegio Oficial de Peritos e Ingenieros Técnicos Industriales de Sevilla.

### EXPERIENCIA PROFESIONAL

— Desde noviembre de 1978 en el ejercicio libre de la profesión (colegiado núm. 5.893), habiendo realizado proyectos de diferentes tipos de industrias y actividades.

— Desde marzo de 1982 hasta abril de 1984, como asalariado de la empresa «LA UNION, S. C. L.», de Villaverde del Río (Sevilla), dedicada a las instalaciones en Media y Baja Tensión.

— Desde octubre de 1986, profesor colaborador del Area de Ingeniería Eléctrica en la Escuela Universitaria Politécnica de la Universidad de Sevilla. ■

## ELECCIONES A CARGOS DEL CONSEJO Y SU COMISION PERMANENTE

El día 14 de marzo de 1987 a las 10.00 horas, se celebraron elecciones para cubrir los cargos de Presidente, Vicepresidente y Secretario del Consejo General, así como de los vocales de su Comisión Permanente. Dado que previamente diversos candidatos habían presentado su renuncia a presentarse a las elecciones convocadas, de acuerdo con los artículos 45, 46 y 48 del Reglamento General de Elecciones, quedaron proclamados electos los siguientes candidatos.

Presidente: **D. Florentino Maña Turbi**

Vicepresidente: **D. José Carlos de Santiago Quintela**

Secretario: **D. Pedro Eugenio Robledo Campo**

Vocales - decanos: **D. Benicio Alonso Pérez, D. Francisco Garzón Cuevas, D. José M.<sup>a</sup> Alonso Pedreira**

La Comisión Permanente tomó posesión de sus cargos el día 10 de abril de 1987 en el Consejo General Extraordinario, convocado para tal efecto.